

# PROJET DE CALCUL AUX ELEMENTS FINIS

## LOGICIEL CASTEM

CALCUL DE LA FORCE MAXIMALE ADMISSIBLE PAR UNE  
TABLE DE CLASSE AVANT SA RUPTURE



## INTRODUCTION

### Objectifs généraux du projet :

- Résoudre un problème de résistance des matériaux par calcul aux éléments finis à l'aide du logiciel CASTEM.
- Apprendre à utiliser le logiciel CASTEM.

### Description générale du projet :

Nous avons décidé de nous intéresser aux déformations que les tables de nos salles de cours subissent, lorsque quelqu'un s'assied dessus.

En effet, nous avons pu voir durant notre scolarité que ces tables cassaient très facilement, (notamment lorsqu'elles ont à supporter le poids d'une personne).

### Objectifs spécifiques au projet :

- Calculer les déformations subies par une table des salles de cours de l'ISABTP, lorsque quelqu'un s'assied dessus.
- Déterminer la force maximale admissible par une table, avant sa rupture.

## HYPOTHESES

Nous avons pris un certain nombre d'hypothèses sur les matériaux constituant la table, sur sa géométrie, ainsi que sur les forces appliquées.

### HYPOTHÈSES SUR LES MATÉRIAUX ET LA GÉOMÉTRIE DE LA TABLE

#### 1) La table :

Elle est composée d'une table en panneaux de fibres de bois, dont les caractéristiques sont les suivantes :

Masse volumique des panneaux :  $\rho = 450 \text{ kg/m}^3$

Épaisseur du panneau : 2,3 cm.

Contrainte de traction : 1,15 MPa.

Module d'élasticité : 11 000 MPa.

Coefficient de poisson : 0,45.

#### 2) Les pieds :

Ce sont des tubes cylindriques en acier de 40mm et 25mm de diamètre extérieur et de 2mm d'épaisseur, soudés entre eux.

Masse volumique de l'acier :  $\rho = 7800 \text{ kg/m}^3$

Module d'élasticité : 21 0 000 MPa.

Coefficient de poisson : 0,25.

#### 3) Assemblage des pieds à la table :

L'assemblage des pieds à la table se fait par l'intermédiaire de six vis (3 de chaque côté), que nous modéliserons par des liaisons de type « articulation » .

## MODELISATION DU PROBLEME SUR CASTEM (cf. « projet1 »)

OPTI DIME 3 ELEM QUA4;  
densite 0.04;

\* ++++++  
\* +modelisation+  
\* ++++++

\*Définition des points :

P1=0. 0. 0.700;  
P2=0. 0.33 0.700;  
P3=0. 0.40 0.700;  
P4=0. 0.47 0.700;  
P5=0. 0.60 0.700;

\*Définition des vis :

V1=0.09 0.47 0.70;  
V2=0.09 0.40 0.70;  
V3=0.09 0.33 0.70;  
V4=1.26 0.47 0.70;  
V5=1.26 0.40 0.70;  
V6=1.26 0.33 0.70;  
V7=0.675 0.47 0.70;  
vtot=V1 et V2 et V3 et V4 et V5 et V6 et V7  
coul jaune;

\*Définition des lignes de construction :

d1= P1 d P2 d P3 d P4 d P5;  
vec1=0.09 0. 0.;  
vec2=0.585 0. 0.;

\*Définition de la surface de la table

surf1=d1 trans vec1 trans vec2 trans vec2  
trans vec1;  
table=vtot et surf1;

\*Définition des points de contact des vis sur  
la face  
inférieure de la table

w1=surf1 point proch V1;  
w2=surf1 point proch V2;  
w3=surf1 point proch V3;  
w4=surf1 point proch V4;  
w5=surf1 point proch V5;  
w6=surf1 point proch V6;  
w7=surf1 point proch V7;  
V8=V7 plus (0. -0.4 0.);

W8=surf1 point proch V8;  
dw13=w1 d w3;  
dw46=w4 d w6;  
dw14=w1 d w4;

\*Définition de l'interface tube acier/table

T2=0.09 0.27 0.70;  
vect1=0. 0.24 0.;  
T1=T2 plus vect1;  
ligne1= T1 droi T2 coul vert;

\*tracé du tube montant incliné en appui  
simple sur le sol

T3=0.09 0.127 0.557;  
vect2=0. -0.04 -0.557;  
vect4=0. -0.0349731 -0.487;  
T4=T3 plus vect2;  
T8=T3 PLUS vect4;  
ligne3= T3 d T8 d T4;

\*tracer de la courbe du montant reliée à  
tube1 (ou tube2)

C1=0.09 0.2856 0.555;  
ligne2=cerc T2 C1 T3;

\*tracé du tube constituant la traverse  
intermédiaire (Φ=2.5)

vect3=1.17 0. 0.;  
T5=T3 plus vect3;  
ligne6= T3 droi T5;

\*tracé du tube constituant la traverse haute  
(Φ=25mm)

T6=0.09 0.47 0.70;  
vect3=1.17 0. 0.;  
T7=T6 plus vect3;  
ligne7= T6 droi V7 d T7;

\*tracé du tube courbé reliant les appuis  
(Φ=40mm)

vect5=0. 0.43 0.;  
T9=T8 PLUS vect5;  
vect7=0. 0.215 -2.0131;



G1=T8 PLUS vect7;  
ligne4= T8 cerc G1 T9;

\*modélisation du petit pied, servant d'appui au sol

vect6=0. 0. -0.07;  
T10=T9 PLUS vect6;  
ligne5= T9 droi T10;

\*Construction de l'autre partie de la table par symétrie

s1=0.675 0.0 0.0;  
s2=0.675 0.69 0.0;  
s3=0.675 0.69 -0.30;  
syme1=ligne5 et ligne4 et ligne2 et ligne3 et ligne1 coul vert;  
syme2= syme1 syme 'PLAN' s1 s2 s3;

T11=T1 plus vect3;  
T21=T2 plus vect3;  
T41=T4 plus vect3;  
T31=T3 plus vect3;  
T81=T8 plus vect3;  
T91=T9 plus vect3;  
G11=G1 plus vect3;  
T101=T10 plus vect3;  
C11=C1 plus vect3;  
ligne31= T31 droi T81 droi T41;  
ligne51= T91 droi T101;  
ligne41= T81 cerc G11 T91;  
ligne11= T11 droi T21;  
ligne21=cerc T21 C11 T31;

lignetot=syme1 et ligne11 et ligne21 et ligne7 et ligne6  
et ligne41 et ligne51 et ligne31 coul vert;  
elim lignetot;

tout=lignetot et table;  
\*formation des groupes d'éléments ayant les mêmes caractéristiques

tube1=ligne1 et ligne2 et ligne3 et ligne5 et ligne11 et ligne21 et ligne31 et ligne51;  
tube2=ligne6 et ligne7;  
hoblon=ligne4 et ligne41;

\*modèle mécanique pour chaque type de barre

motube1=mode tube1 mecanique elastique poutre;  
motube2=mode tube2 mecanique elastique poutre;  
mohoblon=mode hoblon mecanique elastique poutre;  
modall=mode surf1 mecanique elastique coq4;

motot=motube1 et motube2 et mohoblon et modall;

matube1=mate motube1 youn 21.E10 nu 0.25 sect (4901.E-7)  
inry(373.E-9) inrz(373.E-9)  
tors(746.E-9) rho 7800;

\*Caractéristiques mécaniques des différents éléments

matube2=mate motube2 youn 21.E10 nu 0.25 sect (3016.E-7)  
inry(8701.E-11) inrz(8701.E-11)  
tors(174.E-9) rho 7800;

mahoblon=mate mohoblon youn 21.E10 nu 0.25 sect (4901.E-7)  
inry(373.E-9) inrz(373.E-9)  
tors(746.E-9) rho 7800;

madall=mate modall epai 0.023 youn 11.E9 nu 0.45 rho 450;

matot=matube1 et matube2 et mahoblon et madall;

\* ++++++  
\*Matrice de rigidité  
\* ++++++

rig1=rigi motot matot;

\*Poids propre de l'objet

mass1=mass motot matot;  
CHG=manu chpo tout uz -9.81;  
FG=mass1\*chg;

\*On applique un chargement égal à 1KN



AMPF=-1000;  
F1=force Fz AMPF w8;

```
* ++++++
* +conditions limites+
* ++++++
```

\*Définition des différents relâchements et blocage de la structure

rel1=rela ense Ux Uy Uz (V1 et w1);  
rel2=rela ense Ux Uy Uz (V2 et w2);  
rel3=rela ense Ux Uy Uz (V3 et w3);  
rel4=rela ense Ux Uy Uz (V4 et w4);  
rel5=rela ense Ux Uy Uz (V5 et w5);  
rel6=rela ense Ux Uy Uz (V6 et w6);  
rel7=rela ense Ux Uy Uz (V7 et w7);

liaison1=rela ense Uz (ligne1 et dw13);  
liaison2=rela ense Uz (ligne11 et dw46);  
liaison3=rela ense Uz (ligne7 et dw14);

BLO1=BLOQ (T4 et T41 ) Uz;  
BLO2=BLOQ (T101 et T10) Uz;

\*On élimine les nœuds en excès :  
elim lignetot;

RITOT=RIG1 ET BLO1 et BLO2 et rel1 et rel2 et rel3 et rel4 et rel5 et rel6 et rel7 et liaison1 et liaison2 et liaison3;

```
* ++++++
* +resolution+
* ++++++
```

ftot=f1 et fg;  
DEP1=RESOU RITOT ftot;  
def1=defo dep1 tout;  
sig1=sigma dep1 motot matot;

trace tout;

```
* ++++++
* +FORCES AUX NOEUDS+
* ++++++
```

Re=reac ritot dep1;  
list Re;

R1=extr Re Fz w1;  
MESS 'REACTION EN w1' R1 'N';  
R2=extr Re Fz w2;  
MESS 'REACTION EN w2' R2 'N';  
R3=extr Re Fz w3;  
MESS 'REACTION EN w3' R3 'N';  
R4=extr Re Fz w4;  
MESS 'REACTION EN w4' R4 'N';  
R5=extr Re Fz w5;  
MESS 'REACTION EN w5' R5 'N';  
R6=extr Re Fz w6;  
MESS 'REACTION EN w6' R6 'N';  
R7=extr Re Fz w7;  
MESS 'REACTION EN w7' R7 'N';  
R8=extr R1 Fz v3;  
MESS 'REACTION EN v3' R8 'N';


```
*
+++++
* +flèche du point d'application de la
force+
*
+++++
```

DEPG=extr dep1 uz w8;  
DEPG=1.E3\*DEPG;  
MESS 'FLECHE EN w8' DEPG 'mm';

OEIL=0. -1000. 0.;  
OEIL1=-1000. -1000. 500.;  
trac OEIL sig1 motot matot def1;  
trac OEIL1 sig1 motot matot def1;



**RÉSULTAT OBTENU AVEC LE LOGICIEL CASTEM**




Déformée et contraintes de VONMISES

Une force de 1 KN appliquée sur la table entraîne des sollicitations maximales au niveau des vis N°3 et 6. Elle entraîne une force de 1433,9 N dans le panneau de fibres de bois, au niveau de ces vis.



## CALCUL DE LA FORCE MAXIMALE ADMISSIBLE

La rupture se fait au niveau des vis.  
Elle se matérialise par le cône de rupture suivant :



**Surface du cône :**

$$S = \pi \times r \times (r+a)$$

$$r = H \times \tan(45^\circ)$$

$$a = H / (\cos(45^\circ))$$

Après mesure in situ ( sur une table cassée), nous avons :

$$r = 1,5 \text{ cm}$$

Application numérique :

$$S = 1,7065 \times 10^{(-3)} \text{ m}^2$$

Afin de déterminer la force maximale admissible nous allons suivre la démarche suivante :

1. il s'agit de repérer la vis autour de laquelle la force est maximale. Après le calcul effectué par le logiciel CASTEM, nous voyons qu'il

s'agit de la vis V3 (et son symétrique V6).

2. En appliquant une force de 1KN à 7cm du bord de la table, nous obtenons une force de 1433,9 N dans le panneau de fibres de bois.

3. Nous savons que :

$$\tau_G = \sigma_e / 2$$

$\tau_G$  : contrainte de cisaillement.

$\sigma_e$  : limite élastique du panneau = 1,15 Mpa.

**Fmax** : Force maximale que peut supporter le panneau.

Application numérique :

$$\tau_G = 575 \text{ KPa}$$

4. Calculons Fmax :

$$F_{\max} = \tau_G \times S$$

Application numérique :

$$F_{\max} = 981,2375 \text{ N}$$



L'application de  $F = 1 \text{ KN}$  donne une force  $F_1 = 1433,9 \text{ N}$  au niveau de la vis la plus sollicitée. Nous devons également prendre en compte le poids propre du panneau.

5. Poids du panneau :

$$P = e \times L \times l \times \rho \times g$$

Application numérique :  
(Cf. voir hypothèses pour les valeurs)

$$P = 0,023 \times 1,35 \times 0,6 \times 450 \times 9,81$$

$$P = 82,24 \text{ N}$$

6. Déterminons la force à appliquer  $F_2$  sur la table pour arriver à atteindre  $F_{\max}$  au

niveau de la vis, (sachant que la relation liant les efforts appliqués aux efforts engendrés est linéaire) :

$$F_2 = \frac{F_{\max} \times (F + P) - P \times F_1}{F_1}$$

Application numérique :

$$F_2 = 658,35 \text{ N}$$

Nous avons pu vérifier la validité de ce calcul en appliquant des forces  $F$  de différentes amplitudes ( les résultats sont alors tout à fait cohérents) ;

## INTERPRÉTATION DES RÉSULTATS


Une force de 658,35 N, appliquée à 7 cm du bord de la table suffit à engendrer la rupture de la table.

Sachant qu'il s'agit de mobilier scolaire, cette force nous paraît très faible, elle peut être atteinte par le poids d'une personne de 68 Kg.

Nous pouvons considérer qu'il s'agit d'un mobilier de mauvaise qualité. Le nombre de tables cassées en témoigne !


## REMARQUES

### 1. remarque sur la modélisation :



*Modélisation « 3D » réelle*

Nous avons réalisé une première modélisation de la table en 3 dimensions (sans utilisation des lignes moyennes). Les liaisons entre les différents éléments (tubes constituant les pieds) nécessitaient un maillage trop difficile à modéliser en langage Gibiane (cf. « projet2 »). Nous avons donc changé de mode de représentation et utilisé les lignes moyennes.



*Modélisation par lignes moyennes*


## 2. remarque sur les conditions limites :

Les conditions de liaison entre la table et les pieds sont des appuis simples et au niveau des vis des rotules. Nous les avons modélisé par ces relations :

rel1=rela ense Ux Uy Uz (v1 et w1);  
 rel2=rela ense Ux Uy Uz (v2 et w2);  
 rel3=rela ense Ux Uy Uz (v3 et w3);  
 rel4=rela ense Ux Uy Uz (v4 et w4);  
 rel5=rela ense Ux Uy Uz (v5 et w5);  
 rel6=rela ense Ux Uy Uz (v6 et w6);  
 rel7=rela ense Ux Uy Uz (v7 et w7);

liaison1=rela maxi Uz lig1 - Uz dwt1t2;  
 liaison2=rela maxi Uz ligne11 - Uz dwt11t21;  
 liaison3=rela maxi Uz ligne7 - Uz dw14;

Avec cette écriture, nous n'arrivons pas à obtenir les valeurs des forces résultantes au niveau des vis où l'arrachement se produit. (cf. fichier « projet3 »)



## CONCLUSION

- Les difficultés rencontrées :

Apprendre à utiliser le langage *Gibiane* qui permet de communiquer avec le programme. La syntaxe est basée sur l'utilisation de directives, d'opérateurs et de procédures qui ne nous ont pas toujours paru faciles à maîtriser.

Choisir une modélisation et des hypothèses simplificatrices pour pouvoir réaliser le projet dans le temps impartis.

- Les apports de ce projet :

Ce projet nous a permis d'appliquer la méthode de calculs aux éléments finis à un exemple concret, qui nous a paru d'autant plus intéressant qu'il s'agissait d'une structure (la table) que nous utilisons tous les jours.

Nous avons pu découvrir la puissance du logiciel de calcul qu'est *CASTEM*, et réaliser qu'il s'agit d'un outil précieux et extrêmement capable aux multiples applications propres à chaque utilisateur.